

```

Name      Adam3a;
Partno    Not_assigned;
Date      11/5/89;
Revision  3;
Designer  Mark Gordon;
Company    Micro Innovations;
Assembly  Adam_IF_board;
Location  U11;

```

```

/*****
/* This PAL performs all address/IO decoding for the SASI/RS-232/ */
/* Parallel Printer Interface board that goes into the Coleco ADAM*/
/* Computer */
/*****
/* Allowable Target Device Types: PAL16L8 */
/*****
/* This version moves the parallel printer output port to address */
/* 40H. The board has been modified to use the chip select line, */
/* re-inverted, for the printer strobe output. The PAL output */
/* that was freed-up will now be used for external memory board */
/* bank select. It's address will be moved to 42H. The SASI */
/* status input port is now dual addressed at 01H and at 40H (so */
/* that the printer driver routine can look at that port for */
/* printer status). The board has also been re-designed so that */
/* the printer status lines are in different bit locations on the */
/* printer status input port (port 40H). */

```

/** Inputs **/

```

Pin 1      = A7;
Pin 2      = A3;
Pin 3      = !RD;
Pin 4      = !WR;
Pin 5      = A6;
Pin 6      = A5;
Pin 7      = A4;
Pin 8      = A0;
Pin 9      = A1;
Pin 11     = !IORQ;
Pin 13     = A2;

```

/** Outputs **/

```

Pin 12     = !2681_CS ;      /* ADDRESS 2681 CHIP (DUAL RS232) */
Pin 14     = !MEM_STB ;      /* External Memory Board STROBE */
Pin 15     = !SND_CONT ;     /* SEND SASI BUS CONTROL LINES */
Pin 16     = !SND_DATA ;     /* SEND SASI BUS DATA LINES */
Pin 17     = !RD_DATA ;      /* READ SASI BUS DATA LINES */
Pin 18     = !PTR_DATA ;     /* SEND PRINTER DATA PORT */
Pin 19     = !RD_STAT ;      /* READ SASI BUS STATUS LINES */

```

/** Declarations and Intermediate Variable Definitions **/

```

FIELD IO_ADDR = [A7..A0] ;    /* NAME THE IO ADDRESS LINES AS */
                               /* A FIELD */

```

/** Logic Equations **/

```

RD_DATA    = IORQ & RD & IO_ADDR:[0] ; /* SASI READ DATA PORT AT ADDR 0 */
SND_DATA    = IORQ & WR & IO_ADDR:[0] ; /* SASI SEND DATA PORT AT ADDR 0 */
RD_STAT     = (IORQ & RD & IO_ADDR:[1]) # (IORQ & RD & IO_ADDR:[40]) ;

```

```

/* SASI STATUS PORT AT ADDR 01 */
/* AND AT ADDR 40H */
SND_CONT = IORQ & WR & IO_ADDR:[1] ; /* SASI CONTROL PORT AT ADDR 01 */
PTR_DATA = IORQ & WR & IO_ADDR:[40] ; /* PRINTER DATA PORT AT ADDR 40 */
MEM_STB = IORQ & WR & IO_ADDR:[42] ; /* MEMORY BOARD BANK SWITCH PORT */
/* STROBE BY WRITING TO PORT 42 */
2681_CS = IORQ & IO_ADDR:[10..1F] ; /* 2681 AT PORTS 10 THRU 1F */

```